A Short Introduction to Functional Programming

Jan Kuper
j.kuper@utwente.nl

April 19, 2015

Preliminary remark. This text introduces elementary concepts of a functional programming language, and discusses some examples. In functional programmaing part of the module *Programming Paradigms* we will use the functional programming language *Haskell*, see `http://www.haskell.org`.

We remark that *no* text on a programming language will be sufficient to actucally learn to program in that language. For that one has to participate in the practical sessions and do the exercises offered there oneself.

GHCi. In this course we will use the interactive Haskell environment GHCi (for: *Glasgow Haskell Compiler – interactive mode*), to be downloaded from

`https://www.haskell.org/platform/`

The first line of a Haskell file has the structure:

module `<ProgramName>` **where**

followed by your definitions. `ProgramName` should start with a capital letter and be the same as the file name, where the file has the extension "`.hs`"

A Haskell file can be opened in the Haskell interpreter which can be started with the command

`ghci <FileName>`.

The program itself is a plain text file. In `ghci` expressions using functions defined in your program can be evaluated. Apart from that there are various commands, a few practical ones being:

– `:l <FileName>` : to *load* a new Haskell file into `ghci`,

– `:r` : to *reload* the file already loaded,

– `:t <Expression>` : to ask the *type* of the expression,

– `:i <Identifier>` : to get *information* on the definition of the identifier,

– `:h`, `:?` : to get an overview of the available commands.

When a line in a Haskell file contains the symbol `--`, the rest of that line is considered as *comment*. Multi-line comments may be included in `{-` and `-}`.

Further information. Although with programming *the learning is in the doing*, there nevertheless are many good and extensive books and tutorials, some of which are available on-line, see tutorials and books. In particular we mention:

– *Learn You a Haskell for Great Good*, `http://learnyouahaskell.com`,

– *Real World Haskell*, `http://book.realworldhaskell.org`.

Finally, we mention some websites for practical use:

– `http://www.haskell.org/hoogle` : "google-for-haskell", i.e., a search engine for Haskell definitions,

– `https://tryhaskell.org/` : a site to execute simple Haskell expressions on-line.

1 Expressions, definitions

The activity of programming in a functional programming language consists of defining functions, and evaluating their effect on given arguments.

Expressions. We start with a warning: a functional programming language *only* has expressions, there are *no* statements as in imperative or object-oriented languages. In particular, there is no assignment statement. A functional program thus does not "work" by explicitly storing and changing values in computer memory, but by *evaluating* expressions, in order to calculate their values.

As usual, expressions are built up from identifiers, constants, variables, and operations. *The* basic operation in a functional language is *application* of a function `f` to an argument `a`, written as `f a`. That is, the operation of function application is denoted by juxtaposition.

We stress that application (often called "function application") is an operation just like many other well-known operations, like `+`, `*` for operations on numerical values, `>`, `>=` for comparisons, etc. Function application has priority over all other operations, for example: `f 3+5` means `(f 3)+5` and not `f (3+5)`. Parentheses are used to overrule the standard priority rules.

In appendix A a list of standard operations available in Haskell is mentioned (see the documentation on `www.haskell.org` for more)[1].

Definitions. To define a as a name for a certain value, include in the script file a definition of the form

$$a = \cdots$$

[1]It's worthwhile to check these, it can save you a lot of trouble in defining already existing functions.

where "$\cdots$" stands for some expression. The following definition is a very simple example

$$a = 25 + 17$$

Functions are also defined in a standard way. For example, a polynomial function like

$$f(x) = 3x^2 + 5x - 4$$

is defined by the line

$$f\ \ x = 3 * x\hat{\ }2 + 5 * x - 4$$

in a script file. Thus, the formulation is almost exactly the same as the mathematical formulation, except that operations (such als multiplication and exponentiation) all have to be written explicitly, and (as already mentioned above) brackets around the argument of a function are often (but not always) omitted.

After loading this script file into the evaluator, expressions like $f\ 7$, corresponding to the mathematical equivalent $f(7)$, can be evaluated. As expected, the *actual argument* 7 is *substituted* for the *formal parameter* x in the definition of f, after which the prescribed calculation is performed.

In definitions we have the possibility of *pattern matching*: in the definition of a function the formal parameter may – by its form alone – determine which argument fits into the parameter. For example, let g be defined by a definition consisting of three *clauses*:

$$\begin{array}{lcl} g\ 0 & = & 10 \\ g\ 1 & = & 20 \\ g\ x & = & 3 * x \end{array}$$

When g is used in some concrete context, the patterns 0, 1, x in this definition determine which clause is chosen. For example, a call $g\ 1$ will choose the second clause, whereas a call $g5$ will choose the third clause. Naturally, only the actual argument 0 fits in pattern 0, likewise for 1. Since x is a variable, every actual argument fits in pattern x. Clauses are tried in-order (from the first line downwards), so changing the order of the clauses in this definition may make a certain clause unreachable.

An alternative formulation of g, using *guards*, is:

$$\begin{array}{lll} g\ x & \mid x == 0 & = 10 \\ & \mid x == 1 & = 20 \\ & \mid \textbf{otherwise} & = 3 * x \end{array}$$

Mixtures of these two forms are also possible:

$$\begin{array}{lll} g\ x & \mid x == 0 & = 10 \\ & \mid x == 1 & = 20 \\ g\ x & = 3 * x & \end{array}$$

Here, if neither of the guards $x == 0$ and $x == 1$ evaluates to $True$, GHC continues with the final clause.

Where clauses. Definitions may contain `where` clauses, as in

$$\begin{array}{ll} h\ x & = y + y + y \\ & \textbf{where} \\ & \quad y = x\hat{\ }2 \end{array}$$

Clearly, the application $h\ 5$ evaluates to 75.

The definition $y = x\hat{\ }2$ in the where clause is only valid inside the definition of h. Thus, y can not be used from outside this definition, i.e. y is not "visible" from outside the definition. Would we write

$$\begin{array}{lll} h\ 1 & = y + y + y & \textbf{where}\ y = 10 \\ h\ x & = y + y + y & \textbf{where}\ y = x\hat{\ }2 \end{array}$$

the y in the second clause is completely unrelated to the y in the first, i.e. we could have given them different names altogether.

A where clause is important for readability and for efficiency: it assures that $x\hat{\ }2$ is evaluated only once, whereas using the functionally equivalent definition

$$h\ x = x\hat{\ }2 + x\hat{\ }2 + x\hat{\ }2$$

the same subexpression would be evaluated three times[2].

Lay-out. The *lay-out* of definitions is important for Haskell to recognize where a definition begins and ends: the first character of a definition determines the top-left corner of a block that is ended by a character in the same column. Subsequent lines that are part of the same definition should

[2]Of course, the compiler will try and eliminate common subexpressions, but it can not always do this.

start to the right of this character. For example, consider the following two definitions of the function h:

$$
\begin{aligned}
&h\ x \quad = y + z \\
&\qquad\quad \textbf{where} \\
&\qquad\qquad y = 3 * x \\
&z \ \ = 5 * x
\end{aligned}
$$

and

$$
\begin{aligned}
&h\ x \quad = y + z \\
&\qquad\quad \textbf{where} \\
&\qquad\qquad y = 3 * x \\
&\qquad\qquad z = 5 * x
\end{aligned}
$$

Only in the second variant the definition of z is part of the **where** clause, whereas in the first variant it is not. Hence, in the second variant, z can not be used by another definition, whereas in the first variant it can.

In the last definition there are three blocks: h marks the top-left corner of a block containing the whole definition. The keyword **where** creates the possibility for nested definitions in which local names are defined. In the nested definitions, the position of y marks the top-left corner of a nested block, which is ended by the definition of z, since z is in the same column as y. Likewise, z starts a new block as well.

There is another aspect illustrated by this example: in the definition of z the variable x is used. In the second variant above this x refers to the formal parametr x in the expression $h\ x$ at the beginning of the definition. However, since in the first variant above the definition of z is outside the scope of this *introduction* of the variable x, this variable is not valid anymore. Hence, in the first variant the definition of z uses a variable x that must be introduced *outside* the definition of h as well.

Referential transparancy. As in mathematical expressions the value of a variable is the same at all occurrences of that variable inside the clause of a definition, no matter how many lines a clause contains. This is different from imperative programming languages, where the value of a variable may depend on the place where it occurs. Referential transparancy is an important feature when proving that a program is correct and to enable advanced program transformations by the compiler.

2 Types

Every (correct) expression e has a *type* T, written as

$$e :: T$$

A type can be seen as a *set of values* on which certain *operations* are defined. Then $e :: T$ means that the value of e belongs to the set T, and all operations defined on T may be applied to e.

Basic types. There are six basic types in Haskell:

Bool	for truth values (*True* and *False*)
Char	for character values (`'a'`, `'A'`, `'7'`, `'*'`), including non-printable characters, such as tab (`'\t'`) and escape (`'\esc'`).
Int	for 32-bit *signed* integer values (everything between -2147483648, or $-(2^{31})$, and 2147483647, or $2^{31}-1$)
Integer	for arbitrarily large integer values
Float	for 32-bit floating point values (3.1415927)
Double	for 64-bit, i.e. double precision, floating point values (3.141592653589793)

Types may be combined into *compound* types. In this section we discuss three ways to combine types: *tuple types*, *list types*, and *function types*. Later on we discuss other possibilities to define new types (section 5).

Haskell can deal with *type variables* which denote arbitrary types. A type variable is an identifier starting with a lower case letter. Names for specific types should start with a capital letter.

Tuple types. Let a, b, c be arbitrary types, then (a, b), (a, b, c), etc, are the types of *n-tuples*. Expressions of these types are written (respectively) as (x, y), (x, y, z). Thus,

$$(x, y) :: (a, b)$$
$$(x, y, z) :: (a, b, c)$$

where $x :: a$, $y :: b$, etc. Expressions of this form may be used as patterns.

There are two standard *projection* functions *fst* and *snd* in Haskell, which select the first and the second element of a 2-tuple (respectively)[3]. Thus:

$$\mathit{fst}\ (x, y) = x$$
$$\mathit{snd}\ (x, y) = y$$

[3] Haskell has no standard projection functions for tuples of more than two arguments.

List types. Let a be an arbitrary type, then $[a]$ is the type of all *lists* of elements of type a. Examples of list expressions of type $[a]$ are $[\,]$ (the empty list), and $[x, y, z]$ (where x, y and z must all have the same type a). Note that the empty list $[\,]$ has type $[a]$ for *every* type a, i.e., $[\,]$ is *polymorphic*.

Some more examples: $[1, 2, 3, 4, 5, 6]$ is a list of numbers, $[True, True, False]$ is a list of booleans, $[(1, 2), (3, 4), (5, 6)]$ is a list of pairs of numbers.

Lists may have any length (including infinity), and they are *ordered* from left to right. The same element may occur more than once in a list.

The primitive operation for lists is ":" by which an *element* can be added to the front-end of a list: $x : [y, z] = [x, y, z]$. In fact, $[x, y, z]$ is shorthand notation for $x : y : z : [\,]$, i.e. lists are constructed by ":", starting from the empty list[4].

Given an index i and a list xs, the expression $xs!!i$ denotes the i-th element of the list xs[5]. The first element of a list has index 0. Of course, i should be smaller than the length of xs, but not negative.

Lists can be defined by means of *list comprehension*, a set-theory-like way to denote lists. For example, if xs is a list of numbers, then

$$[\ x\hat{\ }2 \mid x <\!- xs\ ,\ x > 0\ ]$$

is the list of squares of all positive numbers from xs. The expression "$x < -xs$" is called a *generator*, and "$x > 0$" a *qualifier*. There may be several of them in a list comprehension, and they are evaluated from left to right (exercise: check the value of $[\ (x, y) \mid x <\!- xs\ ,\ y <\!- ys\ ]$ for some lists xs and ys).

Generators may use patterns. For example, when ps is a list of pairs of type (Int, Int), then

$$[\ x + y \mid (x, y)\ <\!- ps\ ]$$

is the list of all totals of all pairs in ps.

Lists of characters are called *strings*. There is a special notation for strings, for example, the list `['a','Z','8','+','@','?']` may be written as `"aZ8+@?"`.

A final remark about a specific notation for lists:

$$\begin{aligned} [1..5] &= [1, 2, 3, 4, 5] \\ [10, 7..0] &= [10, 7, 4, 1] \\ [1..] &= [1, 2, 3, 4, 5, ... \end{aligned}$$

[4]As shown in this example, ":" is right-associative, i.e. $x : y : [\,]$ is the same as $x : (y : [\,])$

[5]The notation "xs" is mnemonic for lists: it denotes English plural (of "x"), and is pronounced as "x-es."

The last expression denotes an infinite list.

This notation works also for type *Char* (in general for all types in the class *Enum*):

$$\begin{array}{lcl} \texttt{['a'..'e']} & = & \texttt{"abcde"} \\ \texttt{['k','i'..'a']} & = & \texttt{"kigeca"} \end{array}$$

List expressions of the form $[\,]$, $[x, y, z]$, $x : xs$ and "abc", may be used as patterns in function definitions.

There are two basic functions for lists: *head* and *tail*. They yield the first element and everything but the first element of the list respectively. That is:

$$\begin{aligned} head\ (x : xs) &= x \\ tail\ (x : xs) &= xs \end{aligned}$$

Comparable functionas are *last* and *init*, which yield the last element of a list, and the hole lest except the last element (respectively). All four functions give an error when applied to the empty list.

Some further standard operations and functions for lists are:

$$\begin{aligned} [1, 2] +\!\!+ [4, 5, 6] &= [1, 2, 4, 5, 6] \\ length\ [5, 2, 5, 3] &= 4 \\ reverse\ [1, 2, 3, 4] &= [4, 3, 2, 1] \end{aligned}$$

Lists may be defined *recursively*, i.e. the defined list may be used in the definition itself. For example:

$$ones = 1 : ones$$

yields the infinite list

$$[1, 1, 1, ...$$

We remark that because of lazy evaluation (see section 7) it is possible to include infinite lists into your program. For example,

$$head\ ones\ \ =\ \ 1$$

Here, the list *ones* will be evaluated only as far as necessary.

Function types. The type of *functions* from arguments of type a to results of type b is written as $a \mathbin{-\!\!>} b$. If $f :: a \mathbin{-\!\!>} b$ and $x :: a$, then $f\ x :: b$. So, for some standard functions mentioned above we have:

$$\begin{aligned} fst &:: (a, b) \mathbin{-\!\!>}\ a \\ snd &:: (a, b) \mathbin{-\!\!>}\ b \\ head &:: [a] \mathbin{-\!\!>}\ a \\ tail &:: [a] \mathbin{-\!\!>}\ [a] \end{aligned}$$

lstlisting Note that these functions are *polymorphic*, they work for all types a, b.

Type synonyms. Synonyms for structured types may be introduced by using the keyword **type**. A first example from the standard Prelude is

$$\textbf{type}\ String = [Char]$$

A somewhat more elaborated example is

$$\textbf{type}\ Persons = [(String, Int)]$$

which denotes the type of lists of persons, where a person is (implicitly) supposed to be represented by (e.g.) his/her name (*String*) and age (*Int*). Equivalently, we might have written

$$\begin{aligned} \textbf{type}\ Person &= (String, Int) \\ \textbf{type}\ Persons &= [Person] \end{aligned}$$

3 More on functions

Recursion. Functions may be defined *recursively*, i.e. the function to be defined may be used in the definition itself. The standard example of a recursive function definition is the factorial function[6] :

$$\begin{array}{l} fac :: Int \mathbin{-\!\!>} Int \\ fac\ 0 = 1 \\ fac\ n = n * fac\,(n-1) \end{array}$$

Note that applying *fac* to a negative number will lead to non-termination.

[6]Since types are of great help in debugging a program, it is good practice to start a function definition with the type of the function.

One can look at recursive definitions in two ways: *operationally* and *equationally*. The first way imagines all the computational steps which are performed when calculating a specific application. For example, imagine the actual calculation of $fac\ 3$ in your mind, and compute the intermediate results *fac* 0, *fac* 1, *fac* 2 by multiplying the previous result with the corresponding number. This approach is preferably left to the computer.

The second way of looking is based on the observation that the factorial of n simply *is equal to* the product of n and the factorial of $n-1$. Since in the definition of *fac* above we have that

- there is a clause for the base case 0,
- in the recursive clause the argument $n-1$ on the right hand side is "simpler" than n on the left hand side,

and since we may *assume* that *fac* $(n-1)$ yields the correct result (compare the *induction hypothesis* from proofs by induction), we conclude that *fac* n yields the correct result as well[7]. Hence, we may leave the actual *execution* of the computation to the computer, and the programmer may restrict himself to looking at the definition as a set of *equations*.

Functions on lists may also be defined recursively. For example, the already mentioned function *length*, which computes the length of a list (i.e. the number of elements in it), can be defined as follows:

$$
\begin{array}{ll}
\multicolumn{2}{l}{length :: [a] \mathrel{-}\!\!> Int} \\
length\ [\,] & = 0 \\
length\ (x : xs) & = 1 + length\ xs
\end{array}
$$

As before, the equational reading of this definition is comparable with a proof by mathematical induction:

- the *base clause* is the definition for the empty list,
- the *induction hypothesis* says that we may assume that the definition works correctly for a list xs (say for a list of length n),
- the *recursive clause* then defines the function *length* for a list x:xs (i.e., a list of length n+1) as being one more than the length of the list xs (using the induction hypothesis concerning the correctness of the function *length* for xs).

[7] The branch of mathematics that deals with properties like this is called *Recursion Theory*.

Clearly, the above case of the factorial function and the length function is straightforward. However, we remark that in more involved cases of recursive definitions this *equational interpretation* is a great help in getting the definition correct. On the other hand, the *operational interpretation* often is a burden in designing a recursive function.

Currying. In daily (i.e., mathematical) practice, definitions of functions with more than one argument (say three) usually take the form

$$f(x, y, z) = \cdots$$

Using n-tuples and pattern matching, such a definition can be directly translated into a functional language:

$$f\ (x, y, z) = ...$$

Notice that the type of such an f (for an appropriate choice of types a, b, c, d) is

$$f :: (a, b, c) \text{->}\ d$$

In functional programming, functions are often *curried* (after the logician Haskell B. Curry, who is also the language's namesake). The above schema then gets the form (abusing the name f – in fact this is another function, though the result for the same values of x, y, z may be the same):

$$f\ x\ y\ z\ =\ \cdots$$

and the type of f then is:

$$f :: a \text{->}\ b \text{->}\ c \text{->}\ d$$

In $f\ (x, y, z)$ the function f is applied to a tuple of three arguments at once. On the other hand, in $f\ x\ y\ z$ the function f is *first* applied to x, yielding a *function*, which then is applied to y, which in turn still has to be applied to z in order to yield a result of type d. Note, however, that also d may still be a function type.

More formally, function application is *left associative*, i.e. $f\ x\ y\ z$ means $((f\ x)\ y)\ z$. Because of currying, functions can be applied to fewer arguments than one would expect, e.g. the expression $f\ x\ y$ denotes a *function* which still has to be applied to an argument z before yielding a result of type d.

Correspondingly, "$->$" is *right* associative, i.e. $a -> b -> c -> d$ stands for $a -> (b -> (c -> d))$. So, for f we have (let $x :: a$, $y :: b$, $z :: c$)

$$\begin{array}{ll} f & :: a -> b -> c -> d \\ f\ x & :: b -> c -> d \\ f\ x\ y & :: c -> d \\ f\ x\ y\ z & :: d \end{array}$$

For example, define the function *add* as

$$\begin{array}{l} add\ :: Int -> Int -> Int \\ add\ x\ y\ =\ x + y \end{array}$$

Now,

$$g = add\ 3$$

is the function which adds 3 to its argument, e.g. $g\ 39 = 42$. It follows that

$$g :: Int -> Int$$

Equivalently, we might have defined g by the following equation:

$$g\ x = add\ 3\ x$$

However, the first definition is on a higher, and more "abstract" level than the second, and expresses more clearly that a function is an object (or value) in itself.

Sectioning. Functions are written in prefix notation. However, binary functions may be "infixed" by writing it between backwards single back-quotes (‘). For example, with the function *add* from above, this gives

$$39\ \text{‘}add\text{‘}\ 3\ =\ 42$$

Conversely, binary infix operations may be turned into functions (in prefix notation) by *sectioning*, written by parentheses. Consider the expression $3 + 5$. Now, $(3+)$, $(+5)$, $(+)$ are *functions* and the following expressions all evaluate to $3 + 5$:

$$\begin{array}{l} (3+)\ 5 \\ (+5)\ 3 \\ (+)\ 3\ 5 \end{array}$$

A special case is $-$, which denotes both a unary operation (negation) and a binary operation (subtraction). To disambiguate between these two, Haskell's default choice is the binary operation.

Lambda abstraction. Consider the function definition (in mathematical notation)

$$f(x) = 3x^2 + 5x - 4$$

An equivalent definition, again expressing more clearly that a function is an object in its own right, is by so-called *lambda abstraction*[8]:

$$f = \lambda x.\, 3x^2 + 5x - 4$$

Applying f to an argument, e.g. 2, then is evaluated as follows:

$$\begin{aligned} f\ (2) &= (\lambda x.\, 3x^2 + 5x - 4)\ (2) \\ &= 3 * 2^2 + 5 * 2 - 4 \\ &= 18 \end{aligned}$$

Here, the important step is the substitution of 2 for x in the second line. Clearly, this is exactly what we do intuitively when we have to calculate $f(2)$ using the first definition of f above. The advantage of lambda abstraction is that we can use functions in expressions without first introducing names for them.

Lambda abstraction is also possible in Haskell, though the syntax differs slightly[9]. The above lambda term is expressed as

$$f \ = \ \backslash x \text{ -> } 3 * x\hat{\ }2 + 5 * x - 4$$

The variable x is called the *formal parameter* of the lambda term, the part on the right hand side of the arrow is the *body* of the lambda term, the variable x in the body is *bound* by the introduction of x on the left hand side of the arrow. A variable in the body which is not bound by the lambda abstraction is said to be *free*.

A lambda term may have a pattern as formal parameter, for example

$$f \ = \ \backslash(x, y) \text{ -> } x + y$$

defines a function which adds the two elements of an ordered pair. The type of f is

$$f :: (Int, Int) \text{ -> } Int$$

[8]In the 1930's the logician Alonzo Church founded the *lambda calculus* as a formal theory about functions. It may be considered as an abstract formulation of the essence of a functional programming language.

[9]GHC does allow unicode input, but it was specifically chosen not to represent lambda abstraction with a λ, because Greek programmers would lose a letter for their variable names.

To avoid parentheses, it is agreed that the body of a lambda term is extended to the right as far as possible. Thus, in the following example, all parentheses may be left out without changing the meaning:

$$(\backslash x \mathrel{-\!\!>} (\backslash y \mathrel{-\!\!>} (x + y)))$$

However, when used, some brackets may be necessary:

$$(\backslash x \mathrel{-\!\!>} \backslash y \mathrel{-\!\!>} x + y)\ 2\ 3$$

Note that the type of the outer lambda term is

$$Int \mathrel{-\!\!>} Int \mathrel{-\!\!>} Int$$

and the value of this last term is 5. Finally, Haskell allows a shorthand for this lambda abstraction, by allowing a single lambda to bind multiple variables. In other words, the above lambda abstraction may also be written as

$$(\backslash x\ y \mathrel{-\!\!>} x + y)\ 2\ 3$$

Higher order functions. Because of currying, functions can have functions as results. There can also be functions which have functions as *parameters.* Furthermore, there can be *lists* of functions. One says that functions are "first class citizens." Functions with functions as parameters or as results, are called *higher order functions.* Two important higher order functions are *map* and *filter* (more are mentioned in appendix A):

$$\begin{array}{l} map :: (a \mathrel{-\!\!>} b) \mathrel{-\!\!>} [a] \mathrel{-\!\!>} [b] \\ map\ f\ xs\ =\ [\ f\ x \mid x \leftarrow xs\ ] \end{array}$$

That is, *map* applies the function f to all elements of the list xs. For example, two equivalent formulations of adding 5 to all elements of a list are

$$\begin{array}{lcl} map\ (add\ 5)\ [1, 7, 3, 10] & = & [6, 12, 8, 15] \\ map\ (\backslash x \mathrel{-\!\!>} x + 5)\ [1, 7, 3, 10] & = & [6, 12, 8, 15] \end{array}$$

The first uses the curried function *add*, defined before, the second uses lambda abstraction.

For *filter* we have:

$$\begin{array}{l} filter :: (a \mathrel{-\!\!>} Bool) \mathrel{-\!\!>} [a] \mathrel{-\!\!>} [a] \\ filter\ p\ xs\ =\ [\ x \mid x \leftarrow xs\ ,\ p\ x\ ] \end{array}$$

That is, *filter* yields the list of all elements of xs which satisfy property p, i.e. for which $p\ x = True$. For example, two equivalent formulations of selecting all numbers greater than 5 from a list are

$$\begin{array}{ll} filter\ (>5)\ [1,7,3,10] & = [7,10] \\ filter\ (\backslash x \to x > 5)\ [1,7,3,10] & = [7,10] \end{array}$$

The first uses sectioning, the second uses lambda abstraction.

Function composition. A very important higher order operation for functions is *function composition*, denoted (in mathematics) by $\circ$. Its mathematical definition is as follows:

$$(g \circ f)\,(x) = g\,(f\,(x)).$$

Function composition also exists in Haskell, denoted as ".", and defined exactly the same:

$$(g.f)\,x = g\,(f\ x)$$

Since function composition operates on functions without first having to apply these functions to an argument, it offers the possibility to formulate definitions on a higher level of abstraction.

Here is an example to illustrate this. In the example a function *revsort* is defined which sorts a list of pairs by looking at the second element of a pair. That is: a pair of numbers (x, y) "comes before" (notation $((x, y) \leq (x', y'))$ a pair (x', y') if $y < y'$, or, in case $y = y'$, if $x < x'$. In all other cases we have that $(x', y') \leq (x, y)$.

The standard relation $\leq$ of Haskell of Haskell is exactly the other way around: Haskell first looks at the first element of a pair to sort a list of pairs. The definition of *revsort* below will use the standard function *sort* of Haskell, so we first have to define a function *swap*, which reverses the order of the elements in a pair:

$$swap\ (x, y) = (y, x)$$

Now the function *revsort* that sorts a list of pairs according to the above ordering relation (i.e. looking at the second element first) by first swapping all pairs in the list, then sorting the list by the standard function *sort* of Haskell, and then swapping all pairs back again. Swapping every pair in a list of pairs is done by the function (!) *map swap*, and swapping each pair back again is done by the same function *map swap*. The following definition expresses this:

$$revsort = map\ swap\ .\ sort\ .\ map\ swap$$

Note that the definition of *revsort* is purely functional, without mentioning an argument for *revsort*. To see how this works, consider the following example:

$$\begin{array}{lll} \multicolumn{3}{l}{revsort\ [(3,5),(3,6),(3,2)]} \\ (1) & \Rightarrow & (map\ swap\ .\ sort\ .\ map\ swap)\ [(3,5),(3,6),(3,2)] \\ (2) & \Rightarrow & map\ swap\ (sort\ (map\ swap\ [(3,5),(3,6),(3,2)])) \\ (3) & \Rightarrow & map\ swap\ (sort\ [(5,3),(6,3),(2,3)]) \\ (4) & \Rightarrow & map\ swap\ [(2,3),(5,3),(6,3)] \\ (5) & \Rightarrow & [(3,2),(3,5),(3,6)] \end{array}$$

Here, on step (1) the definition of *revsort* is applied, and on step (2) the definition of function composition is used. Note that *map swap* is considered here as a *function*, being the result of applying the function *map* to the function *swap*. On steps (3) and (5) the function *swap* is applied to all elements in the list, and on step (4) the list is sorted by the predefined Haskell function *sort*, using the standard ordering on tuples.

Note that the resulting list indeed is sorted on the second elements of its tuples.

Operations and functions. Both operations (+, :, ., etc) and functions (*head*, *sort*, etc) can be applied to arguments to yield some result. However, even though we may turn operations into functions (by sectioning), and binary functions into operations (by back quoting), there are important differences between the two. First of all, a function is a *value* of some (function) type, and an operation is not. To check this, compare the reactions of GHCi to "`:t +`" and "`:t (+)`". Likewise for "`:t div`" and "`` :t `div` ``".

In general one might say that an operation "glues" expressions together into a bigger expression and determines how the resulting expression has to be evaluated. Remember that also between a function and its argument there is an operation: *function application*, though that operation is not written explicitly.

In Haskell we may not only define functions, but also operations, including higher order operations. For example, function composition is predefined in Haskell, but we might have defined it ourselves as

$$f\ .\ g\ =\ \backslash x \mathrel{-\!\!>} f\ (g\ x)$$

As an example to combine several of the above issues, we write a function that selects from a list of 2-tuples of numbers those pairs whose first element is even, and those pairs whose second element is between 5 and 10, using only one mentioning of *filter*.

First we define two operations that combine properties (i.e. boolean valued functions):

$$
\begin{aligned}
p \;\&\; q &= \backslash x \rightarrow p\ x\ \&\&\ q\ x \\
p \;\#\; q &= \backslash x \rightarrow p\ x\ ||\ q\ x
\end{aligned}
$$

Thus, a value x has *property* p & q, if x has both property p *and* property q. Likewise, x has *property* $p \# q$, if x has property p *or* property q (or both).

Now consider the expression:

$$filter\ (((==0)\ .\ (\text{`}mod\text{`}\ 2)\ .\ fst)\ \#\ (((>5)\ .\ snd)\ \&\ ((<10)\ .\ snd)))$$

To understand this expression, first consider the subexpression:

$$(==0)\ .\ (\text{`}mod\text{`}\ 2)\ .\ fst$$

The function *fst* chooses the first element of a 2-tuple, then the function ‘*mod*‘ 2 determines the remainder after division by 2 (being 0 or 1), and the function ==0 checks whether something is equal to 0. Combining these functions with function composition (“.”), i.e., applying these functions one after another, thus checks whether the first element of a pair is even.

It is left to the reader that the subexpression

$$((>5)\ .\ snd)\ \&\ ((<10)\ .\ snd)$$

is a property which says that the second element of a pair is greater than 5 *and* smaller than 10. Thus, the full expression given above filters from a list of pairs those pairs of which the first element is even *or* the second elemend is between 5 and 10.

Exercise. Analyze what the type of this expression is, design an appropiate testing expression for it, predcit what the result of that expression will be, and evaluate it in GHCi.

Recursors. The previously given examples of recursive function definitions have the form of *primitive recursion*, i.e. a function is defined for a basic value (such as `0`, `[]`), and for the “next” value(s) (`n+1`, `x:xs`)[10]. This

[10] There are other forms of recursion, such as partial recursion, or general recursion.

form of recursion can be expressed in a general way by so-called *recursors*. For natural numbers, the (primitive) recursor `recr` is defined as follows:

```
recr f a  0    = a
recr f a (n+1) = f (n+1) (recr f a n)
```

Note that the definition of `recr` is itself primitive recursive.

Using `recr` the factorial function `fac` can be defined as follows:

```
fac = recr (*) 1
```

Now `fac 4` evaluates to `4*(3*(2*(1*1)))` , which evaluates to `24`. That is to say, the computation is built up from the right. There is also a left variant, `recl`:

```
recl f a  0    = a
recl f a (n+1) = recl f (f a (n+1)) n
```

Check that, since multiplication is associative and commutative, the definition

```
fac = recl (*) 1
```

leads to the same result, though the order of computation is different.

The same principle can be applied to recursion for lists, leading to the important functions `foldr` ("fold-right") and `foldl` ("fold-left")

```
foldr f a []     = a
foldr f a (x:xs) = f x (foldr f a xs)

foldl f a []     = a
foldl f a (x:xs) = foldl f (f a x) xs
```

To illustrate a somewhat tricky difference between the two, consider the following two definitions of the function `length`:

```
length = foldr g 0   where  g x y = y + 1
length = foldl g 0   where  g x y = x + 1
```

The difference between `foldr` and `foldl` can be shown as follows:

```
foldr f a [b,c,d] = b `f` (c `f` (d `f` a))
foldl f a [b,c,d] = ((a `f` b) `f` c) `f` d
```

This explains the naming "right" and "left".

The variants `foldl1` and `foldr1` take the first and last (respectively) element of the list as starting value, i.e.

```
foldl1 f (x:xs) = foldl f x xs
```

Thus `foldl1`, `foldr1` both give an error message for the empty list.

Exercises. Derive the types of `foldr` and `foldl`.

Define `reverse`, `concat`, `map` and `filter` using `foldr` or `foldl`.

4 Type classes

Many functions are *polymorphic*, they work for any type. For example

```
head :: [a] -> a
map  :: (a->b) -> [a] -> [b]
```

for all types `a`, `b`. Remember that the definitions of `head` and `map` only use the structural properties of list and function types and not information from the types `a`, `b` themselves.

On the other hand, there are functions that work for a specific *class* of types. For example, relational operations such as "`<`" exist for types such as `Int`, `Char`, `[Int]`, but not for function types `a->b`. Types for which an ordering relation exist belong to the class `Ord`, and an ordering relation such as "`<`" has the following type:

```
(<) :: Ord a => a -> a -> Bool
```

This can be read as follows: "`(<)` has the type `a->a->Bool`, assuming that the type `a` belongs to the class `Ord`".

Likewise, the type of the test for equality is as follows:

```
(==) :: Eq a => a -> a -> Bool
```

These operations are called *overloaded* since they exist for several types, but their implementations may differ for the various concrete types. For example, "`<`" is implemented differently for `Int`s and for `Float`s. Assuming that the concrete definitions of "`==`" and "`<`" are given, several other operations and functions can be defined while defining the class `Ord` itself:

```
class Eq a => Ord a where
  x >  y  =  y < x
  x <= y  =  x<y || x==y
  x >= y  =  x>y || x==y
```

The operations "`>`", "`<=`", "`>=`" are called *methods* since they are valid for every type in the class `Ord`. It is left to te reader to extend this definition of the class `Ord` for methods such as `min` and `max`.

The prefix "`Eq a =>`" means that a type `a` only belongs to class `Ord`, if `a` belongs to class `Eq`. Hence, the class `Ord` is a *subclass* of the class `Eq`, where the class `Eq` itself is defined as follows:

```
class Eq a where
  x /= y  =  not (y == x)
  x == y  =  not (y /= x)
```

The methods in class `Eq` seem to be circular, but the definition of either `==` or `/=` for a concrete type will overrule the method definition in the class itself, and thus break the circularity.

If a type belongs to a class, it is called an *instance* of that class. When defining some type `A` (see section 5), a programmer may declare that `A` is an instance of class `Ord` and at the same time define the elementary relation "`<`" for type `A` as follows:

```
instance Ord A where
  x < y  =  ...
```

All other methods of the class `Ord` now automatically are valid for type `A` as well.

Remember that since `Ord` is a subclass of `Eq`, type `A` should first be declared to be an instance of class `Eq` before `A` will really belong to class `Ord`.

In several cases, methods can be derived by Haskell, and the programmer need not define them himself. For that the keyword "`deriving`" can be used. We will see examples of that below.

There exist many predefined classes in Haskell. Here we only mention the type `Show` of all types whose values can be transformed into strings such that they can be shown on an ascii screen. Most types belong to the class `Show`, but functions typically do not. This class has as a method the function

```
show :: Show a => a -> String
```

which transforms a value of any type `a` into a printable string, whenever type `a` belongs to the class `Show`.

5 Type definitions.

The above mentioned possibility of type synonyms does not introduce new types with new values, it only gives a name to types which could be constructed from given types already. In this section we discuss two possibilities to define new types whose values are also new and explicitly defined by the programmer: *algebraic types* and *record types*. These types can be defined by using the keyword `data`. Newly defined types have to be given a name, called a *type constructor*.

Algebraic types. Values of *Algebraic Data Types* (ADTs) are created by so-called *data constructors*. The basic example is the type `Bool`, which is defined as

```
data Bool = True | False
```

Thus, "`Bool`" is a *type* constructor, whereas "`True`" and "`False`" are *data* constructors. Constructors start with a capital letter. The symbol "`|`" stands for "or".

More generally, constructors may take types as arguments, for example the type

```
data BasicType = I Int | B Bool | C Char
```

contains *values* like `I 5`, `I 42`, `B True`, `C 'a'` and `C '7'`, which all have type `BasicType`. The values of this type might be called "labeled," where the data constructors `I`, `B`, `C` may be seen as the labels. Labeled values may be used as patterns in definitions.

ADTs may be *parameterized*, i.e. also a type constructor may have types as arguments:

```
data PQtype a = P a | Q a
```

For example, the types `PQtype Int` and `PQtype Char` are concrete types of this general form, and we have that

```
P 25    :: PQtype Int
P 'a'   :: PQtype Char
Q 'a'   :: PQtype Char
Q True  :: PQtype Bool
```

We remark that `P 25` and `Q 25` are different values of `PQtype Int`.

As mentioned above, the type `PQtype a` may be declared by the programmer to belong to a given class by using the keyword `instance`. For the class `Eq` this can be done as follows:

```
instance Eq a => Eq (PQtype a) where
   P x == P y   =   x == y
   Q x == Q y   =   x == y
   _   ==   _   =   False
```

Since equality on type `PQtype a` is defined in terms of equality on the types `a`, the type `a` has to be an instance of the class `Eq` as well. Note, however, that the programmer is free to choose an alternative definition for equality.

The same can be done for the classes `Ord` and `Show`:

```
instance Ord a => Ord (PQtype a) where
   P x < P y   =   x < y
   Q x < Q y   =   x < y
   P _ < Q _   =   True
   Q _ < P _   =   False

instance Show a => Show (PQtype a) where
   show (P x)   =   "P␣" ++ show x
   show (Q x)   =   "Q␣" ++ show x
```

Here the programmer chose to let `P`-terms be smaller than `Q`-terms (since in the definition of the type `PQtype a` the data constructor `P` precedes the data constructor `Q`), but of course, here too he might have made another choice.

In practice the above schedule occurs very often, and therefore Haskell has the possibility to derive standard classes already in the definition of the type itself, using the keyword `deriving`:

```
data PQtype a = P a | Q a  deriving (Eq, Ord, Show)
```

Here it is implicitly assumed that the type `a` belongs to the classes `Eq`, `Ord` and `Show`. Haskell's choice for defining `==`, `<` and `show` is as in the above instance declarations.

Recursive types. ADTs may be used recursively, for example

```
data Tree = Leaf Int | Node Char Tree Tree
```

contains a.o. the following values:

```
Leaf 5
Node 'a' (Leaf 3) (Leaf 10)
Node 'b' (Node 'a' (Leaf 3) (Leaf 10)) (Leaf 5)
```

Intuitively, these expressions denote tree-structures (hence the name `Tree`) with numbers at the leaves, and characters at the internal nodes. Furthermore, in trees of type `Tree` every internal node has two subtrees, i.e., the type `Tree` contains binary trees.

In fact, natural numbers (`Nat`) and lists of elements of type `a` (`List a`) might also be defined as recursive types:

```
data Nat    = Zero | Succ Nat
data List a = Nil  | Cons a (List a)
```

Remember that in the final example, the type `a` stands for an arbitrary type. Clearly, the type `List a` is isomorphic to the predefined type `[a]`.

Functions on recursive types may be defined recursively, i.e., by using the recursive structure of the type definition. For example, the length of a list of type `List a` can be defined as follows:

```
length Nil         = 0
length (Cons x xs) = 1 + length xs
```

When the function `length` is used, the variable `x` in the pattern `Cons x xs` binds to a value of type `a`, whereas the variable `xs` binds to a value of type `List a`. Thus, this value of `xs` again is of the form `Nil`, or `Cons y ys`.

Exercise. Define several standard functions on `Nat` and `List a`. For example, define the arithmetical functions (addition, multiplication, etc.) for the type `Nat`, and the list functions (`head`, `reverse`, `map`, etc.) for the type `List a`.

Record types. Remember the type `Person` on page 9, defined as a type synonym for the 2-tuple `(String,Int)`. In that way of defining the type `Person` it is implicit that the string and the integer are supposed to denote the name and the age (resepctively) of the person. An alternative way to define a type for persons in which name and age are explicit, is by defining a *record type* as follows:

```
data Person  = Pers { name::String, age::Int }
```

Thus, this definition consists of the keyword `data`, mentions (`name`, `age`) and their types between curly brackets, and labels it with a data constructor (`Pers`). The data constructor may or may not be the same as the *type* constructor (here `Person`). There is no limitation on the number of fields, a field name may be any identifier (starting with a lower case letter) and types may be chosen freely.

Clearly, the list of persons still can be defined by a type synonym:

```
type Persons = [Person]
```

A concrete record of type `Person` may now be defined as

```
p0 = Pers { name="Bill", age=25 }
```

In the type `Person` the *field names* can extract the values by using them as *functions*:

```
name p0  ==>  "Bill"
age  p0  ==>  25
```

Correspondingly, their types are

```
name :: Person -> String
age  :: Person -> Int
```

Record fields may be given different values by *record updating*:

```
p0{age=35}                 ==>  Pers {name="Bill" , age=35}
p0{age=35, name="Chris"}  ==>  Pers {name="Chris", age=35}
```

Note that the order in which the field names are listed does not matter, though GHC shows them in the order as given in the type definition.

As suggested by the definition of the type `Person`, records indeed are an extension of algebraic data types, and we may include several records within the same type:

```
data RecTypes
      = Employee { address :: String, salary :: Int }
      | Supplier { address :: String, account :: Int }
```

Note that both record variants contain a field with name `address`. That is possible, as long as the content type of the field `address` is the same, in this case `String`, such that its type is well-defined:

```
address :: RecTypes -> String
```

As with algebraic data types, record types may be declared to be instances of certain classes by the keyword `instance`, and as before, several standard classes may be derived by the keyword `deriving`. Thus, after

```
data Person  = Pers { name::String, age::Int }
               deriving (Eq,Ord,Show)
```

values of type `Person` may be compared with equality and ordering relations, and they may be shown on an ascii screen. The result is as expected, with the remark that ordering relations look at the values of the fields in the order that these fields are introduced in the type definition.

Records may be partial and neither order, nor lay-out are important, so

```
p1 = Pers {name="Chris"}
p2 = Pers {age=35, name="Chris"}
```

both are (partial) values of type `Person`.

When the content of filed `age` of record `p1` is needed in a computation, an exception will arise. But as long as it is not needed, there is no problem (for an explanation, see section 7 on "lazy evaluation").

Records (also partially) may be used as *patterns*, for example

```
isAdult (Pers {age=x})  =  x >= 18
```

Note that the *structure* of this record pattern is indicated by the part `Pers {age=...}`, whereas `x` is a variable, i.e. a "nested" pattern which "catches" the value of the corresponding field. The names of these variables can be chosen freely – *including* `age` itself[11]. That is to say, the following definition is equivalent to the above one:

```
isAdult (Pers {age=age})  =  age >= 18
```

6 Modules.

Definitions in a script file may be packed in a module, and modules may be imported in other script files. A module has to start with a line like (below some variants of this format will be discussed)

```
module ModName where
```

followed by the definitions in the module. The keyword `module` starts with a lowercase letter, whereas the name of the module starts with a capital letter. The name of the file containing the module should have the same name as the module itself.

A module may be imported in another file by including the line

```
import ModName
```

after which the functions and types from the module `ModName` may be used.

An important role for modules is to *hide* the definitions of types and/or functions. In order to make functions and types from a module known in a file where the module is imported, they have to be *exported* explicitly from the module by mentioning them in the heading of the module as follows:

```
module ModName ( Type1
               , Type2(A,B)
               , Type3(..)
```

[11] In fact, there is a wealth of possibilities in using patterns and updating fields in Haskell, but for that we refer to the GHC Manual, especially chapter 8. See `http://www.haskell.org/ghc/docs/latest/html/users_guide/index.html`

```
        , fun1
        , fun2
        , ...
        ) where
```

Here, the types `Type1`, `Type2`, `Type3`, and the functions `fun1`, `fun2` are known from outside. However, none of the constructore of `Type1` will be known outside the module, whereas from `Type2` only constructors `A` and `B` will be known. The notation "`..`" means that *all* constructors from `Type3` are exported.

Even though a type or function is exported in this way, it still can be made inaccessible from the script file where the module is imported by saying, e.g.,

```
import ModName hiding fun1
```

This is useful in case the programmer wants to define a function with the same name as `fun1`.

On the othe rhand, it is possible to indicate which function is meant by prefixing it with its module:

```
ModName.fun1
```

indicates the function `fun1` from module `ModName`.

As an example, consider the following module for sets:

```
module Set (Set, empty, add, union, member) where

data Set a = S [a]  deriving Eq

instance Show a => Show (Set a) where
  show (S x) = show x

empty :: Set a
empty = S []

add :: Eq a => a -> Set a -> Set a
add x (S xs) = S (nub (x:xs))

union :: Eq a => Set a -> Set a -> Set a
union (S xs) (S ys) = S (nubp (xs++ys))

member :: Eq a => a -> Set a -> Bool
```

```
member x (S xs) = elem x xs

nub :: Eq a => [a] -> [a]
nub  []    = []
nub (x:xs) = x : nub (filter (/=x) xs)
```

First of all, note that the only way to have access to elements of type `Set a` is by means of the exported functions. In general, a programmer will know their types and their semantics, but not their precise definitions.

Internally in the module the type `Set a` has only one constructor `S` which "packs" lists into sets. Note that the constructor `S` is not exported, so a programmer can not use it. Furthermore, since the function `show` is redefined in such a way that the constructor `S` is not shown, a programmer using this module will not see the constructor `S` at all.

Since sets do not contain duplicates, the function `nub` removes those. Note that the function `nub` is not usable from outside since it is not exported.

7 Lazy evaluation

An expression may be evaluated in various orders. For example, let

```
f x = x * x,
```

and evaluate `f (3+4)`. Then there are three possibilities:

```
f (3+4) => f 7             => 7 * 7     => 49
f (3+4) => (3+4) * (3+4) => 7 * (3+4) => 7 * 7 => 49
f (3+4) => (3+4) * (3+4) => (3+4) * 7 => 7 * 7 => 49
```

The first is called *eager evaluation* (argument first), the second *leftmost-outermost* evaluation. The third is a mixed form and has no name. Since the argument is only evaluated once, the first strategy seems to be more efficient. Nevertheless, Haskell, like many other functional languages, chooses the second possibility (extended with sharing, see below). The advantage is that the final answer will be found if there exists one. For example, define

```
from n = n : from (n+1)
```

Then

```
from 3 = [3,4,5,...
```

Evaluate

```
head (tail (from 3))
```

Then first fully evaluating the argument `from 3` would give an infinite reduction and thus no answer at all. Following the leftmost-outermost strategy one gets the answer `4`:

```
head (tail (from 3))
     =1=>  head (tail (3 : from (3+1)))
     =2=>  head (from (3+1))
     =3=>  head ((3+1) : from ((3+1)+1))
     =4=>  3+1
     =5=>  4
```

Note that an argument is only evaluated insofar it is needed to evaluate an expression which contains the argument as a part. Thus, in step 1, `g 3` has to be evaluated only one step, such that the rule for `tail` can be applied in step 2. In order to apply the rule for `head` (step 4), again `g` has to be applied one step first (step 3). Note that only one addition is performed (step 5).

To repair the possible loss in efficiency by computing the same expression over and over again, *sharing* is applied, i.e. when an argument is substituted for a formal parameter, every evaluation step in one of the copies of the argument is executed in all its copies. In the first example above, this would mean that `(3+4)` would be evaluated only once:

```
f (3+4) => (3+4) * (3+4) => 7 * 7 => 49
```

Leftmost-outermost reduction with sharing is called *lazy evaluation*.

In Haskell, there is one[12] function which gives some control over the order of evaluation: `seq`. The function `seq` first evaluates its first argument, but delivers the second:

```
seq (3+4) f (3+4) => seq 7 f (3+4) => f (3+4) => ...
```

This example only shows the idea of the function `seq`, but not its usefulness.

8 Some mixed topics

Debugging. Haskell has a few standard facilities for debugging, the simplest of which is the module `Debug.Trace`, which must be imported in order to use the function

```
trace :: String -> a -> a
```

An expression like

[12] Actually, GHC has a few more, but they are non-standard.

```
trace a b
```

prints `a` on the standard output, and continues evaluates to `b`.

Error. We conclude this section on functions with mentioning the special polymorphic function

```
error :: String -> a
```

The expression

```
error "this␣is␣an␣error␣message"
```

may be used in any function definition. Evaluating it causes the program to terminate after the error message is printed.

9 FPPrac Package and Graphical Environment

The `fpprac` package that is part of the Haskell Environment for this course exports the following modules:

FPPrac Only exports `FPPrac.Prelude`

FPPrac.Prelude Defines `Number`, a combined integral and floating number type, and corresponding instances of most numeric type classes (`Num`, `Real`, `Integral`, `Fractional`, `RealFrac`, `Floating`). Also defines (non-type class) functions normally found in `Prelude` that either have arguments or a result of a standard Haskell numeric type, but will now use an argument of type `Number` in that position. For example: the function `take` is given type `Number -> [a] -> [a]` instead of the `Int -> [a] -> [a]`. It reexports all the other functions of `Prelude` that do not fall under the above mentioned category.

FPPrac.Graphics Exports the `Graphics.Gloss` module from the `gloss-glfw` package, and defines the `graphicsout` function which can display a value of type `Picture` on the screen.

FPPrac.Events Exports the `Graphics.Gloss.Interface.Game` module from the `gloss-glfw` package, and defines the `installEventHandler` function that passer keyboard/mouse input to a user-defined function, and displays the picture returned by that same function on the screen.

9.1 FPPrac.Prelude

The following functions using a value of type `Number` as either an argument or result are exported by `FPPrac.Prelude`:

- `length :: [a] -> Number` – `length` returns the length of a finite list as a `Number`
- `(!!) :: [a] -> Number -> a` – List index (subscript) operator, starting from 0.
- `replicate :: Number -> a -> [a]` – `replicate n x` is a list of length `n` with `x` the value of every element.
- `take :: Number -> [a] -> [a]` – `take n`, applied to a list `xs`, returns the prefix of `xs` of length `n`, or `xs` itself if `n > length xs`.
- `drop :: Number -> [a] -> [a]` – `drop n`, applied to a list `xs`, returns the suffix of `xs` after the first `n` elements, or `[]` if `n > length xs`.
- `splitAt :: Number -> [a] -> ([a],[a])` – `splitAt n xs` returns a tuple where first element is `xs` prefix of length `n` and second element is the remainder of the list.

9.2 Graphics

9.2.1 graphicsout

The `graphicsout` function, which has type `Picture -> IO ()`, opens a new window and displays the given picture. It should only be called once during the execution of a program, and should be called at the highest point in your function hierarchy! You can use the following commands once the window is open:

Close Window – esc-key

Move Viewport – left-click drag, arrow keys.

Rotate Viewport – right-click drag, control-left-click drag, or homeend-keys.

Zoom Viewport – mouse wheel, or page updown-keys.

9.2.2 Picture type

The `Picture` datatype has the following constructors:

Blank – A blank picture, with nothing in it.

Polygon Path – A polygon filled with a solid color.

Line Path – A line along an arbitrary path.

Circle Float – A circle with the given radius.

ThickCircle Float Float – A circle with the given thickness and radius. If the thickness is 0 then this is equivalent to Circle.

Text String – Some text to draw with a vector font.

Bitmap Int Int ByteString – A bitmap image with a width, height and a ByteString holding the 32 bit RGBA bitmap data.

Color Color Picture – A picture drawn with this color.

Translate Float Float Picture – A picture translated by the given x and y coordinates.

Rotate Float Picture – A picture rotated by the given angle (in degrees).

Scale Float Float Picture – A picture scaled by the given x and y factors.

Pictures [Picture] – A picture consisting of several others.

Where `Path` is a list of points, `[Point]`, and `Point` is a tuple of an x and a y coordinate, `(Float,Float)`. The picture uses a Cartesian coordinate system (`http://en.wikipedia.org/wiki/Cartesian_coordinate_system`), where the floating point value 1.0 is equivalent to the width of 1 pixel on the screen. The window created by either `graphicsOut` or `installEventHandler` is 800 by 600 pixels, so the visible coordinates run from -400 to +400 on the X-axis, and -300 to +300 on the Y-axis. You can enlarge or decrease the visible range by scaling the window. A picture is will have (0,0) as its original center, and if required, will have to be moved to a different position using the `Translate` constructor.

Predefined values of type `Color` are: `black`, `white`, `red`, `green`, `blue`, `yellow`, `cyan`, `magenta`, `rose`, `violet`, `azure`, `aquamarine`, `chartreuse`, `orange`. More information about the `Color` type can be found in the API documentation that is generated when you install the Haskell Environment.

9.3 Events

9.3.1 installEventHandler

The event mode lets you manage your own input. Pressing ESC will still close the window, but you don't get automatic pan and zoom controls like with `graphicsout`. You should only call `installEventHandler` once during the execution of a program, and should be called at the highest point in your function hierarchy! The `installEventHandler` is of type: `forall userState . => String -> (userState -> Input -> (userState,Maybe Picture)) -> userState -> IO ()`. The first argument is the name of the window, the second argument is the event handler that you want to install, and the third argument the initial state of the event handler. Your event handler should be a function that takes to arguments: a self-defined internal state, and a value of type `Input`. It should return an updated state, and a value of `Maybe Picture`. If this value is `Nothing`, the current picture remains on the screen; if this value is `Just <new_picture>`, the picture on the screen will be replaced with the one defined by the contents of `<new_picture>`. Your event-handler will be called 50 times per second, and will be passed a value of `NoInput` if there is no input at that time.

9.3.2 Input type

The `Input` datatype has the following constructors:

NoInput – No input

KeyIn Char – Keyboard key x is pressed down; `' '` for spacebar, `\t` for tab, `\n` for enter

MouseDown (Float,Float) – Left mouse button is pressed at location (x,y)

MouseDown (Float,Float) – Left mouse button is released at location (x,y)

MouseMotion (Float,Float) – Mouse pointer is moved to location (x,y)

Appendix A: Some standard operators and functions

Below some important operations and functions predefined in Haskell. See the on-line help function for more, and for a more extensive description. Definitions can be found under `installation | initialisation` in Haskell-help.

Remark. The list below presents the standard Haskell types, *without* the type `Number` as used in this course, see section 9.1.

```
negate, abs,
signum         :: Num a => a -> a
+, -, *        :: Num a => a -> a -> a
div, mod       :: Integral a => a -> a -> a
/              :: Fractional a => a -> a -> a
^              :: (Num a, Integral b) => a -> b -> a
abs, exp, log,
sqrt, sin, cos Floating a => a -> a
```
various arithmetical operations and functions

```
min, max       :: Ord a => a -> a -> a
```
gives the minimum, maximum of two arguments

```
not            :: Bool -> Bool
&&, ||         :: Bool -> Bool -> Bool
```
boolean operations negation, conjunction, disjunction

```
isLower,
isUpper        :: Char -> Bool
```
says whether a letter is lower-case or upper-case

```
isAlpha        :: Char -> Bool
```
says whether a character is a letter

```
isDigit        :: Char -> Bool
```
says whether a character is a digit

```
isAlphaNum     :: Char -> Bool
```
says whether a character is a letter or a digit

```
ord            :: Char -> Int
```

converts a character to its Unicode number

`chr` :: `Int -> Char`
converts a Unicode number to the corresponding character

`toLower`,
`toUpper` :: `Char -> Char`
converts a letter to lower-case, upper-case

`==`, `/=` :: `Eq a => a -> a -> Bool`
`>`, `>=`,
`<`, `<=` :: `Ord a => a -> a -> Bool`
various comparison operations

`even`, `odd` :: `Integral a => a -> Bool`
says whether a (integral) number is even or odd

`:` :: `a -> [a] -> [a]`
adds element to the front end of a list (*cons*)

`length` :: `[a] -> Int`
length of a list

`!!` :: `[a] -> Int -> a`
list indexing

`++`, `\\` :: `[a] -> [a] -> [a]`
list concatenation, list subtraction

`head`, `last` :: `[a] -> a`
`tail`, `init`,
`reverse` :: `[a] -> [a]`
`elem` :: `Eq a => a -> [a] -> Bool`
tests whether a list contains a given element

`concat` :: `[[a]] -> [a]`
concats a list of lists into one list

`sort` :: `Ord a => [a] -> [a]`
`sum` :: `Num a => [a] -> a`
`minimum`,
`maximum` :: `Ord a => [a] -> a`
`take`, `drop` :: `Int -> [a] -> [a]`

`takeWhile`,
`dropWhile` :: `(a->Bool) -> [a] -> [a]`
various functions on lists

`insert` :: `Ord a => a -> [a] -> [a]`
inserts an element into an ordered list

`and`, `or` :: `[Bool] -> Bool`
yields the conjunction, disjunction of a list of booleans

`lines` :: `String -> [String]`
breaks a string at newlines (`'\n'`) into a list of strings

`unlines` :: `[String] -> String`
glues a list of strings with `'\n'`

`fst` :: `(a,b) -> a`
yields the first element of a pair

`snd` :: `(a,b) -> b`
yields the second element of a pair

`zip` :: `[a] -> [b] -> [(a,b)]`
turns two lists into a list of pairs

`zipWith` :: `(a->b->c) -> [a] -> [b] -> [c]`
zips two lists and applies a function to the corresponding elements

`map` :: `(a->b) -> [a] -> [b]`
applies a function to all elements in a list

`filter` :: `(a->Bool) -> [a] -> [a]`
selects those elements from a list which satisfy a property

`foldl` :: `(a->b->a) -> a -> [b] -> a`
"folds" a list with a function, starting with a given value. Works from left to right through the list

`foldr` :: `(a->b->b) -> b -> [a] -> b`
like `foldl`, but works from right to left

`foldl1`,
`foldr1` :: `(a->a->a) -> [a] -> a`

like `foldl`, `foldr`, with first, last element
of the list as starting value. Error for empty list

`.`	::	`(b->c) -> (a->b) -> (a->c)` function composition
`seq`	::	`a -> b -> b` partially evaluates first argument, and delivers the second
`error`	::	`String -> a` causes error with given string as error message

Literature

There are a few good websites where lots of information on functional programming can be found. For example:

`http://www.haskell.org`
`http://www.cs.kun.nl/~clean`

From these sites both tutorials and implementations can be downloaded. Some books:

BIRD, R., P. WADLER, *Introduction to Functional Programming*, Prentice Hall, London, 1988

BIRD, R., *Introduction to Functional Programming Using Haskell* (second edition), Prentice Hall, London, 1998

DAVIE, A.J.T., *An Introduction to Functional Programming Systems Using Haskell*, Cambridge UP, Cambridge, 1992

HUDAK, P., *The Haskell School of Expression*, Cambridge UP, Cambridge, 2000

HUTTON, G., *Programming in Haskell*, Cambridge UP, Cambridge, 2007

KOOPMAN, P., R. PLASMEIJER, M. VAN EEKELEN, S. SMETSERS, *Functional Programming in Clean*, Nijmegen, 2001 (freely available from `http:\\www.cs.kun.nl/~clean`)

RABHI, F., G. LAPALME, *Algorithms, a Functional Programming Approach*, Addison-Wesley, Harlow, 1999

THOMPSON, S., *Miranda, the Craft of Functional Programming*, Addison-Wesley, Harlow, 1995

THOMPSON, S., *Haskell, the Craft of Functional Programming*, Addison-Wesley, Harlow, 1996